OrbitalFrequencies subpackage of KerrGeodesics

Define usage for public functions

```
BeginPackage["KerrGeodesics`OrbitalFrequencies`",
    {"KerrGeodesics`ConstantsOfMotion`"}];

KerrGeoFrequencies::usage =
 "KerrGeoFrequencies[a, p, e, x] returns the orbital frequencies."

Begin["`Private`"];
```

Roots of the radial and polar equations

```
(* Returns the roots of the radial equation, as given by Fujita and Hikida *)
KerrGeoRadialRoots[a_, p_, e_, x_, En1_: Null, Q1_: Null] :=
 Module[{M = 1, En = En1, Q = Q1, r1, r2, r3, r4, AplusB, AB},
If[En == Null, En = KerrGeoEnergy[a, p, e, x]];
If[Q == Null,  Q = KerrGeoCarterConstant[a, p, e, x]];

r1 = p / (1 - e);
r2 = p / (1 + e);
AplusB = (2 M) / (1 - En^2) - (r1 + r2); (*Eq. (11)*)
AB = (a^2 Q) / ((1 - En^2) r1 r2); (*Eq. (11)*)
r3 = (AplusB + Sqrt[(AplusB)^2 - 4 AB]) / 2; (*Eq. (11)*)
r4 = AB / r3;

{r1, r2, r3, r4}

]
```

This code uses the polar equation (z^2-zm^2)(a^2(1-E0^2)z^2-zp^2)==0 as the Polar equation. Hence zp is a*Sqrt[1-E0^2]*zp in other sources.

```
KerrGeoPolarRoots[a_, p_, e_, x_] := Module[{En, L, Q, zm, zp},
 {En, L, Q} = Values[KerrGeoConstantsOfMotion[a, p, e, x]];
 zm = Sqrt[1 - x^2];
 zp = (a^2 (1 - En^2) + L^2 / (1 - zm^2)) ^ (1 / 2);
 {zp, zm}
]
```

Orbital Frequencies

Orbital frequency calculations from Fujita and Hikida, Class. Quantum Grav .26 (2009) 135002, arXiv:0906.1420

Schwarzschild

```
KerrGeoMinoFrequencies[0 | 0., p_, 0, x_] :=
 <| "\!\(\*SubscriptBox[\(ϒ\), \(r\)]\)" -> Sqrt[((-6 + p) p) / (-3 + p)],
  "\!\(\*SubscriptBox[\(ϒ\), \(θ\)]\)" -> p Sqrt[1 / ((-3 + p) x^2)] x,
  "\!\(\*SubscriptBox[\(ϒ\), \(ϕ\)]\)" -> (p x) / Sqrt[(-3 + p) x^2],
  "Γ" -> Sqrt[p^5 / (-3 + p)] |>;

KerrGeoMinoFrequencies[0 | 0., p_, e_, x_] :=
 <| "\!\(\*SubscriptBox[\(ϒ\), \(r\)]\)" ->
     (Sqrt[-((p (-6 + 2 e + p)) / (3 + e^2 - p))] π) / (2 EllipticK[(4 e) / (-6 + 2 e + p)]),
  "\!\(\*SubscriptBox[\(ϒ\), \(θ\)]\)" -> p / Sqrt[-3 - e^2 + p],
  "\!\(\*SubscriptBox[\(ϒ\), \(ϕ\)]\)" -> (p x) / (Sqrt[-3 - e^2 + p] Abs[x]),
  "Γ" -> 1 / 2 Sqrt[(-4 e^2 + (-2 + p)^2) / (p (-3 - e^2 + p))]
      (8 + 1 / ((-4 + p)^2 EllipticK[(4 e) / (-6 + 2 e + p)])
         (-(((-4 + p) p^2 (-6 + 2 e + p) EllipticE[(4 e) / (-6 + 2 e + p)]) / (-1 + e^2)) +
           (p^2 (28 + 4 e^2 - 12 p + p^2) EllipticK[(4 e) / (-6 + 2 e + p)]) / (-1 + e^2) -
           (2 (6 + 2 e - p) (3 + e^2 - p) p^2 EllipticPi[(2 e (-4 + p)) /
                ((1 + e) (-6 + 2 e + p)), (4 e) / (-6 + 2 e + p)]) / ((-1 + e) (1 + e)^2) +
           (4 (-4 + p) p (2 (1 + e) EllipticK[(4 e) / (-6 + 2 e + p)] + (-6 - 2 e + p)
                 EllipticPi[(2 e (-4 + p)) / ((1 + e) (-6 + 2 e + p)),
                   (4 e) / (-6 + 2 e + p)])) / (1 + e) + 2 (-4 + p)^2
             ((-4 + p) EllipticK[(4 e) / (-6 + 2 e + p)] - ((6 + 2 e - p) p EllipticPi[(16 e) /
                   (12 + 8 e - 4 e^2 - 8 p + p^2), (4 e) / (-6 + 2 e + p)]) / (2 + 2 e - p)))) |>;

KerrGeoBoyerLindquistFrequencies[0 | 0., p_, 0, x_] :=
 <| "\!\(\*SubscriptBox[\(Ω\), \(r\)]\)" -> Sqrt[-6 + p] / p^2,
  "\!\(\*SubscriptBox[\(Ω\), \(θ\)]\)" -> (Sqrt[1 / x^2] x) / p^(3 / 2),
  "\!\(\*SubscriptBox[\(Ω\), \(ϕ\)]\)" -> (p x) / Sqrt[p^5 x^2] |>;

KerrGeoProperFrequencyFactor[0 | 0., p_, 0, x_] := p^2
```

```
KerrGeoProperFrequencyFactor[0 | 0. , p_, e_, x_] :=
 (p^2 ((1 + e) (28 + 4 e^2 + (-12 + p) p) - ((1 + e) (-4 + p) (-6 + 2 e + p)
          EllipticE[(4 e) / (-6 + 2 e + p)] + 2 (6 + 2 e - p) (3 + e^2 - p)
          EllipticPi[(2 e (-4 + p)) / ((1 + e) (-6 + 2 e + p)), (4 e) / (-6 + 2 e + p)]) /
       EllipticK[(4 e) / (-6 + 2 e + p)])) / (2 (-1 + e) (1 + e)^2 (-4 + p)^2)
```

Kerr

```
KerrGeoMinoFrequencies[a_, p_, e_, x_] :=
 Module[{M = 1, En, L, Q, r1, r2, r3, r4, ϵ0, zm, a2zp,
   ϵ0zp, zmOverZp, kr, kθ, ϒr, ϒθ, rp, rm, hr, hp, hm, ϒϕ, Γ},
{En, L, Q} = Values[KerrGeoConstantsOfMotion[a, p, e, x]];

{r1, r2, r3, r4} = KerrGeoRadialRoots[a, p, e, x, En, Q];
ϵ0 = a^2 (1 - En^2) / L^2;
zm = 1 - x^2;
a2zp = (L^2 + a^2 (-1 + En^2) (-1 + zm)) / ( (-1 + En^2) (-1 + zm));

ϵ0zp = -((L^2 + a^2 (-1 + En^2) (-1 + zm)) / (L^2 (-1 + zm)));

(*zmOverZp=
    If[a==0,0,zm/((L^2+a^2 (-1+En^2) (-1+zm))/(a^2 (-1+En^2) (-1+zm)))];*)
zmOverZp = zm / ((L^2 + a^2 (-1 + En^2) (-1 + zm)) / (a^2 (-1 + En^2) (-1 + zm)));


kr = Sqrt[(r1 - r2) / (r1 - r3) (r3 - r4) / (r2 - r4)]; (*Eq.(13)*)
kθ = Sqrt[zmOverZp]; (*Eq.(13)*)
ϒr = (Pi Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)]) / (2 EllipticK[kr^2]);
  (*Eq.(15)*)
ϒθ = (Pi L Sqrt[ϵ0zp]) / (2 EllipticK[kθ^2]); (*Eq.(15)*)

rp = M + Sqrt[M^2 - a^2];
rm = M - Sqrt[M^2 - a^2];
hr = (r1 - r2) / (r1 - r3);
hp = ((r1 - r2) (r3 - rp)) / ((r1 - r3) (r2 - rp));
hm = ((r1 - r2) (r3 - rm)) / ((r1 - r3) (r2 - rm));

(*Eq. (21)*)
ϒϕ = (2 ϒθ) / (Pi Sqrt[ϵ0zp]) EllipticPi[zm, kθ^2] +
     (2 a ϒr) / (Pi (rp - rm) Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)])
      ((2 M En rp - a L) / (r3 - rp) (EllipticK[kr^2] -
           (r2 - r3) / (r2 - rp) EllipticPi[hp, kr^2]) - (2 M En rm - a L) / (r3 - rm)
         (EllipticK[kr^2] - (r2 - r3) / (r2 - rm) EllipticPi[hm, kr^2]));
```

```
Γ = 4 M^2 En + (2 a2zp En ϒθ) / (Pi L Sqrt[ϵ0zp]) (EllipticK[kθ^2] - EllipticE[kθ^2]) +
     (2 ϒr) / (Pi Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)])
      (En / 2 ((r3 (r1 + r2 + r3) - r1 r2) EllipticK[kr^2] + (r2 - r3) (r1 + r2 + r3 + r4)
             EllipticPi[hr, kr^2] + (r1 - r3) (r2 - r4) EllipticE[kr^2]) +
        2 M En (r3 EllipticK[kr^2] + (r2 - r3) EllipticPi[hr, kr^2]) +
        (2 M) / (rp - rm) (((4 M^2 En - a L) rp - 2 M a^2 En) / (r3 - rp)
             (EllipticK[kr^2] - (r2 - r3) / (r2 - rp) EllipticPi[hp, kr^2]) -
            ((4 M^2 En - a L) rm - 2 M a^2 En) / (r3 - rm)
             (EllipticK[kr^2] - (r2 - r3) / (r2 - rm) EllipticPi[hm, kr^2])));

<| "\!\(\*SubscriptBox[\(ϒ\), \(r\)]\)" -> ϒr,
  "\!\(\*SubscriptBox[\(ϒ\), \(θ\)]\)" -> Abs[ϒθ],
  "\!\(\*SubscriptBox[\(ϒ\), \(ϕ\)]\)" -> ϒϕ,
  "Γ" -> Γ |>

]

KerrGeoMinoFrequencies[(1 | 1.), p_, e_, x_] :=
 Module[{M = 1, a = 1, En, L, Q, r1, r2, r3, r4, ϵ0, zm,
   a2zp, ϵ0zp, zmOverZp, kr, kθ, ϒr, ϒθ, rp, rm, hr, hM, ϒϕ, Γ},

{En, L, Q} = Values[KerrGeoConstantsOfMotion[a, p, e, x]];

{r1, r2, r3, r4} = KerrGeoRadialRoots[a, p, e, x, En, Q];
ϵ0 = a^2 (1 - En^2) / L^2;
zm = 1 - x^2;
a2zp = (L^2 + a^2 (-1 + En^2) (-1 + zm)) / ((-1 + En^2) (-1 + zm));

ϵ0zp = -((L^2 + a^2 (-1 + En^2) (-1 + zm)) / (L^2 (-1 + zm)));

(*zmOverZp=
    If[a==0,0,zm/((L^2+a^2 (-1+En^2) (-1+zm))/(a^2 (-1+En^2) (-1+zm)))];*)
zmOverZp = zm / ((L^2 + a^2 (-1 + En^2) (-1 + zm)) / (a^2 (-1 + En^2) (-1 + zm)));


kr = Sqrt[(r1 - r2) / (r1 - r3) (r3 - r4) / (r2 - r4)]; (*Eq.(13)*)
kθ = Sqrt[zmOverZp]; (*Eq.(13)*)
ϒr = (Pi Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)]) / (2 EllipticK[kr^2]);
  (*Eq.(15)*)
ϒθ = (Pi L Sqrt[ϵ0zp]) / (2 EllipticK[kθ^2]); (*Eq.(15)*)

hM = ((r1 - r2) (r3 - M)) / ((r1 - r3) (r2 - M));
```

```
hr = (r1 - r2) / (r1 - r3);

(*ϒϕ and Γ from Appendix B for a=M case*)

ϒϕ = (2 ϒθ) / (π Sqrt[ϵ0zp]) EllipticPi[zm, kθ^2] +
     (2 a ϒr) / (π Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)])
      ((2 M En) / (r3 - M) (EllipticK[kr^2] - (r2 - r3) / (r2 - M) EllipticPi[hM, kr^2]) +
        (2 M^2 En - a L) / (2 (r3 - M)^2) ((2 - ((r1 - r3) (r2 - r3)) / ((r1 - M) (r2 - M)))
             EllipticK[kr^2] + ((r1 - r3) (r2 - r4) (r3 - M)) / ((r1 - M) (r2 - M) (r4 - M))
             EllipticE[kr^2] + (r2 - r3) / (r2 - M) ((r1 - r3) / (r1 - M) +
               (r2 - r3) / (r2 - M) + (r4 - r3) / (r4 - M) - 4) EllipticPi[hM, kr^2]));

Γ = 4 M^2 En + (2 a^2 En a2zp ϒθ) / (π L Sqrt[ϵ0zp]) (EllipticK[kθ^2] - EllipticE[kθ^2]) +
     (2 ϒr) / (π Sqrt[(1 - En^2) (r1 - r3) (r2 - r4)])
      (En / 2 ((r3 (r1 + r2 + r3) - r1 r2) EllipticK[kr^2] + (r2 - r3) (r1 + r2 + r3 + r4)
             EllipticPi[hr, kr^2] + (r1 - r3) (r2 - r4) EllipticE[kr^2]) +
        2 M En (r3 EllipticK[kr^2] + (r2 - r3) EllipticPi[hr, kr^2]) +
        (2 M (4 M^2 En - a L)) / (r3 - M)
         (EllipticK[kr^2] - (r2 - r3) / (r2 - M) EllipticPi[hM, kr^2]) +
        (M^2 (2 M^2 En - a L )) / (r3 - M)^2 ((2 - ((r1 - r3) (r2 - r3)) / ((r1 - M) (r2 - M)))
             EllipticK[kr^2] + ((r1 - r3) (r2 - r4) (r3 - M)) / ((r1 - M) (r2 - M) (r4 - M))
             EllipticE[kr^2] + (r2 - r3) / (r2 - M) ((r1 - r3) / (r1 - M) +
               (r2 - r3) / (r2 - M) + (r4 - r3) / (r4 - M) - 4) EllipticPi[hM, kr^2]));

<| "\!\(\*SubscriptBox[\(ϒ\), \(r\)]\)" -> ϒr,
 "\!\(\*SubscriptBox[\(ϒ\), \(θ\)]\)" -> Abs[ϒθ],
 "\!\(\*SubscriptBox[\(ϒ\), \(ϕ\)]\)" -> ϒϕ,
 "Γ" -> Γ |>
]

KerrGeoBoyerLindquistFrequencies[a_, p_, e_, x_] := Module[{ϒr, ϒθ, ϒϕ, Γ},

{ϒr, ϒθ, ϒϕ, Γ} = Values[KerrGeoMinoFrequencies[a, p, e, x]];

 <| "\!\(\*SubscriptBox[\(Ω\), \(r\)]\)" -> ϒr,
  "\!\(\*SubscriptBox[\(Ω\), \(θ\)]\)" -> ϒθ,
  "\!\(\*SubscriptBox[\(Ω\), \(ϕ\)]\)" -> ϒϕ
 |> / Γ
]
```

```
KerrGeoProperFrequencyFactor[a_, p_, e_, x_] :=
Module[{ρ1, ρ2, ρ3, ρ4, zm, zp, T},
    {ρ1, ρ2, ρ3, ρ4} = KerrGeoRadialRoots[a, p, e, x];
    {zp, zm} = KerrGeoPolarRoots[a, p, e, x];
    T = KerrGeoEnergy[a, p, e, x];
    With[{kr = (ρ1 - ρ2) / (ρ1 - ρ3) (ρ3 - ρ4) / (ρ2 - ρ4),
     kθ = a^2 (1 - T^2) (zm / zp)^2, hr = (ρ1 - ρ2) / (ρ1 - ρ3)},

        1 / 2 (- ((2 zp^2) / (-1 + T^2)) + ρ1 (-ρ2 + ρ3) + ρ3 (ρ2 + ρ3))
         + ((ρ1 - ρ3) (ρ2 - ρ4) EllipticE[kr]) / (2 EllipticK[kr])
         + (zp^2 EllipticE[kθ]) / ((-1 + T^2) EllipticK[kθ]) +
     ((ρ2 - ρ3) (ρ1 + ρ2 + ρ3 + ρ4) EllipticPi[hr, kr]) / (2 EllipticK[kr])
    ]
]

KerrGeoProperFrequencies[a_, p_, e_, x_] := Module[{MinoFreqs, P},
    MinoFreqs = KerrGeoMinoFrequencies[a, p, e, x];
    P = KerrGeoProperFrequencyFactor[a, p, e, x];
    <|"\!\(\*SubscriptBox[\(ω\), \(r\)]\)" ->
     MinoFreqs["\!\(\*SubscriptBox[\(ϒ\), \(r\)]\)"] / P,
   "\!\(\*SubscriptBox[\(ω\), \(θ\)]\)" ->
     MinoFreqs["\!\(\*SubscriptBox[\(ϒ\), \(θ\)]\)"] / P,
   "\!\(\*SubscriptBox[\(ω\), \(ϕ\)]\)" ->
     MinoFreqs["\!\(\*SubscriptBox[\(ϒ\), \(ϕ\)]\)"] / P |>
]
```

Generic function for choosing between frequencies w.r.t different time coordinates

```
Options[KerrGeoFrequencies] = {"Time" -> "BoyerLindquist"}
SyntaxInformation[KerrGeoFrequencies] =
  {"ArgumentsPattern" -> {_, _, _, _, OptionsPattern[]}};
KerrGeoFrequencies[a_, p_, e_, x_, OptionsPattern[]] :=
 Module[{M = 1, En, L, Q, r1, r2, r3, r4, ϵ0, zm, a2zp,
    ϵ0zp, zmOverZp, kr, kθ, ϒr, ϒθ, rp, rm, hr, hp, hm, ϒϕ, Γ},


If[OptionValue["Time"] == "Mino",
   Return[KerrGeoMinoFrequencies[a, p, e, x][[1 ;; 3]]]];

If[OptionValue["Time"] == "BoyerLindquist",
   Return[KerrGeoBoyerLindquistFrequencies[a, p, e, x]]];

If[OptionValue["Time"] == "Proper",
   Return[KerrGeoProperFrequencies[a, p, e, x][[1 ;; 3]]]];

]
```

Close the package

```
End[];

EndPackage[];
```